Beginning Coq Reference Sheet

Example Proof

```
(* “for all things you could prove, *)
(*    if you have a proof of it, then you have a proof of it.” *)
Theorem my_first_proof : (forall A : Prop, A -> A).
Proof.
  intros A.
  intros proof_of_A.
  exact proof_of_A.
  (* Press C-c C-Enter after the next command to see what the proof *)
  (* would look like in a declarative fashion; i.e., without tactics in λ-calculus. *)
  Show Proof.
  (* Earlier in the proof, this commands shows a partial λ-term. *)
Qed.
```

- As you can see, **every** Coq command ends with a period.
- `Prop` is the type of *propositions*: The type of things which could have a proof.
- Coq uses 3 ‘languages’:
  1. *Vernacular*: The top-level commands that begin with a capital letter.
  2. *Tactics*: Lower-case commands that form the proof; ‘proof strategies’.
  3. *Terms*: The expressions of what we want to prove; e.g., `forall, Prop, ->`.

  This is unsurprising since a language has many tongues.
- *Proofs and functions are the same thing!*
  - We can view what we call a proof as function by using `Show Proof`, as above.
  - We can write functions directly or use [proof] tactics to write functions!

Administrivia, Syntax

- Every Coq command ends with a period.
- The phrase *Theorem T identifying statement S is proven by P* is formalised as

```
Theorem T : S.  (* T is only a name and can be used later. *)
Proof.
P  (* See the current state of the proof in the CoqIde by clicking, in the toolbar,
      on the green arrow pointing at a yellow ball;
      or do "C-c C-Enter" in Proof General with Emacs. *)
Qed.
```

- Instead of `Theorem`, you may also see proofs that start with `Example`, `Lemma`, `Remark`, `Fact`, `Corollary`, and `Proposition`, which all mean the same thing. This difference is mostly a matter of style.
- A defined theorem is essentially a function and so it can be used with arguments, in order to prove a result, as if it were a function.
- The command `Admitted`, in-place of `Qed`, can be used as a placeholder for an incomplete proof or definition.
  - Useful if you have a subgoal that you want to ignore for a while.
- `Abort`, in-place of `Qed`, is used to give up on a proof for the moment, say for presentation purposes, and it may be begun later with no error about theorems having the same name.

Comments `(* I may be a multiline comment. *)`

Stand alone commands As top-level items, we may make commands for:

Normalisation `Compute X` executes all the function calls in `X` and prints the result.

Type inspection Command `Check X.` asks Coq to print the type of expression `X`.

Introduce local definitions Two ways,

- Simple alias: `pose (new_thing := complicated_expression).`
- More involved: Write tactic `assert (x : X).` to define a new identifier `x` for a proof of `X` which then follows, and is conventionally indented.

Imports Loading definitions from a library,

```
Require Import Bool.
```

Local tactic application `t in s` performs the tactic `t` only within the hypothesis, term, `s`. For example, `unfold defnName in H` performs a local rewrite in hypothesis `H`.

- By default, tactics apply to the current subgoal.

`intros` Tactic: ‘∀, ⇒’ Introduction

- To prove $\forall\ x,\ Px$: “Let x be arbitrary, now we aim to prove Px.”
- This strategy is achieved by the `intros x` tactic.
- To prove ∀ `x0 x1 ... xN, Pxs` use `intros x0 x1 ... xN` to obtain the subgoal `Pxs`.
  - Using just “`intros.`” is the same as `intros H H0 H1 ... HN-1.` —‘H’ for hypothesis.
  - Prop names are introduced with the name declared; e.g., “`intros.`” for “∀ `A : Prop, Px`” uses the name `A` automatically.
- Note: `(A → B) = (∀ a:A, B)` and so `intros` works for ‘→’ as well.
- `Show Proof` will desugar `intros` into argument declarations of a function.

`exact` Tactic

- If the goal matches a hypothesis `H` *exactly*, then use tactic `exact H`.
- `Show Proof` desugars `exact H` into `H`, which acts as the result of the currently defined function.

Tactics `refine` & `pose` [local declarations]

If the current goal is C and you have a proof $p : A_0 \to \cdots \to A_n \to C$, then `refine (p _ _ ... _)` introduces n possibly simpler subgoals corresponding to the arguments of `p`.

- This is useful when the arguments may be difficult to prove.
- If we happen to have a proof of any A_i, then we may use it instead of an ‘_’.
- Any one of the underscores could itself be `(q _ ... _)` if we for some proof `q`.

In contrast, you could declare proofs p_i for each A_i, the arguments of `p`, first *then* simply invoke `exact (p p0 p1 ... pN)`. To do this, use the `pose` tactic for forming local declarations: `pose (res := definition_of_p_i)`. The parentheses are important.

- `Show Proof` desugars `pose` into `let...in...` declarations.

Algebraic Datatypes —`Inductive` and `case`

'forall' and type construction allow us to regain many common datatypes, including ∃, ∧, ∨, =, ~, ⊤, ⊥.

The vernacular command `Inductive` lets us create new types.

- After a type, say, `T` is defined, we are automatically provided with an elimination rule `T_rec` and an induction principle `T_ind`.
- Use "`Check T_rec.`" to view their types.

Tactic "`case x.`" creates subgoals for every possible way that `x` could have been constructed —where ideally `x` occurs in the goal.

- In particular, for empty type `False`, it creates no new subgoals.
- If `x` occurs in some hypothesis of interest, then try performing the `case` *before* introducing the hypothesis so that the case analysis propagates into it.
- `case` only changes the goal —never the context.
- Whenever you use this tactic, indent and place `- admit.` for each possible case, so that way you don't forget about them and the indentation make it clear which tactics are associated with which subgoals.
  - Tactic `admit` let's us ignore a goal for a while, but the proof is marked incomplete.
- If `x` is constructed from by `cons a0 ... aN`, then the goal obtains these arguments. It's thus very common to have "`case H. intros.`"; in-fact it's so common that this combination is packaged up as the `destruct` tactic.
  `case H. intros a0 ... aN.` ≈ `destruct H as [a0 ... aN]`.
  - If no `a_i` are provided, the `as` clause may be omitted, and `H`-ypothesis names are generated.
  - If the `case` provides multiple cases, then `destruct` won't work.

If the goal is a value of an ADT, use `refine (name_of_constructor _ ... _)` then build up the constituents one at a time.

- For example, to prove `A ∧ B`, use `refine (conj _ _)`.

Examples of Common Datatypes

- `Prop` Type
  - A `Prop` either has a proof or it does not have a proof.
  - Coq restricts Prop to being either proven or unproven, rather than true or false.
- ℕaturals

```
Inductive nat : Set :=
  | O : nat   (* Capital-letter O, not the number zero. *)
  | S : nat -> nat.
```

- Options

```
Inductive option (A : Type) : Type :=
  | Some : A -> option A
  | None : option A.
```

- Lists

```
Inductive list (A : Type) : Type :=
  | nil : list A
  | cons : A -> list A -> list A.

Infix "::" := cons (at level 60, right associativity) : list_scope.
```

`True, False, true, false`

- The empty Prop, having no proofs, is `False`.
- The top Prop, having a single proof named `I`, is `True`.
- The `bool` type has two values: `true` and `false`.

```
Inductive False : Prop := .

Inductive True : Prop :=
  | I : True.

(* 'Set' is the type of normal datatypes. *)
Inductive bool : Set :=
  | true : bool
  | false : bool.

(* From: Require Import Bool *)
Definition eqb (p q : bool) : bool :=
  match p, q with
    | true, true => true
    | true, false => false
    | false, true => false
    | false, false => true
  end.
```

In the boolean library there is a function `Is_true` which converts booleans into their associated Prop counterparts.

```
(* "Require Import" is the vernacular to load definitions from a library *)
Require Import Bool.
```

Exercises:

```
Theorem two: not (Is_true(eqb false true)). Abort.
Theorem same: forall a : bool, Is_true(eqb a a). Abort.
Theorem ex_falso_quod_libet : (forall A : Prop, False -> A). Abort.
Theorem use_case_carefully: (forall a:bool, (Is_true (eqb a true)) -> (Is_true a)).
```

Notation, Definition, and the tactics `fold` and `unfold`

`Definition` is a vernacular command that says two expressions are interchangeable. Below `(not A)` and `A -> False` are declared interchangeable.

```
Definition not (A:Prop) := A -> False.

Notation "~ x" := (not x) : type_scope.
```

- ◇ A common proof technique is to 'unfold' a definition into familiar operators, work with that, then 'fold' up the result using a definition.
- ◇ Tactics `unfold defnName` and `fold defnName` will interchange them.
- ◇ In Coq, we use the tactic `unfold f` to rewrite the goal using the definition of `f`, then use `fold f`, if need be.
- ◇ `Notation` creates an operator and defines it as an alternate notation for an expression.
- ◇ (Use `intros` when working with negations since they are implications!)

```
(* If this is a recursive function, use 'Fixpoint' in-place of 'Definition'.*)
Definition my_function (a0 : A0) ··· (a99 : A99) : B :=
  match a0 , ..., a99 with
  | C₀ p₀ ... p_n, ..., C_k q₀ ··· q_m => definition_here_for_these_constructors_Cᵢ
  ⋮
  end.
```

Telescoping If `x₀, ···, x_n` have the same type, say `T`, we may declare their typing by `(x₀ ··· x_n : T)`.

Notation Before the final ".", we may include a variant of `where "n + m" := (my_function n m) : B_scope.` for introducing an operator immediately with a function definition.

Function Tactic `simpl` —"simplify"

- ◇ If the current subgoal contains a function call with all its arguments, `simpl` will execute the function on the arguments.
  - ∘ Sometimes a unfold is needed before `simpl` will work.

on application If we have `imp : A -> B`, `a : A` then `imp a` is of type `B`. This also works if the `imp` contains `forall`'s.

Conjunction & Disjunction —products & sums— and 'iff'

```
(* Haskell: Either a b = Left a | Right b *)
Inductive or (A B:Prop) : Prop :=
  | or_introl : A -> A \/ B
  | or_intror : B -> A \/ B
where "A \/ B" := (or A B) : type_scope.

(* Haskell: Pair a b = MkPair a b *)
Inductive and (A B:Prop) : Prop :=
  conj : A -> B -> A /\ B
where "A /\ B" := (and A B) : type_scope.

Definition iff (A B:Prop) := (A -> B) /\ (B -> A).
Notation "A <-> B" := (iff A B) : type_scope.
```

Existence ∃

```
Inductive ex (A:Type) (P:A -> Prop) : Prop :=
  ex_intro : forall x:A, P x -> ex (A:=A) P.

Notation "'exists' x .. y , p" := (ex (fun x => .. (ex (fun y => p)) ..))
  (at level 200, x binder, right associativity,
   format "'[' 'exists' '/ ' x .. y , '/ ' p ']'")
  : type_scope.
```

Note that the constructor takes 3 arguments: The predicate `P`, the witness `x`, and a proof of `P x`.
If we pose a witness beforehand then `refine (ex_intro _ witness _).`, Coq will infer `P` from the current goal and the new subgoal is the proof that the witness satisfies the predicate. This is the way to prove an existence claim.

Searching for Existing Proofs

- ◇ Searching for utility functions, proofs, that involve a particular identifier by using `Search`.
- ◇ In contrast, `SearchPattern` takes a pattern with holes '_' for expressions.
- ◇ Finally, `SearchRewrite` only looks for proofs whose conclusion in an equality involving the given pattern.

```
Search le.
(* le_n: forall n : nat, n <= n *)
(* le_0_n: forall n : nat, 0 <= n *)
(* min_l: forall n m : nat, n <= m -> Nat.min n m = n *)
(* and many more *)

(* Let's load some terribly useful arithmetic proofs. *)
Require Import Arith Omega.

SearchPattern (_+_ <= _+_).
(* plus_le_compat_r: forall n m p : nat, n <= m -> n + p <= m + p *)
(* Nat.add_le_mono: forall n m p q : nat, n <= m -> p <= q -> n + p <= m + q *)
(* etc. *)

SearchRewrite (_ + (_ - _)).
(* le_plus_minus: forall n m : nat, n <= m -> m = n + (m - n) *)
(* le_plus_minus_r: forall n m : nat, n <= m -> n + (m - n) = m *)
(* Nat.add_sub_assoc: forall n m p : nat, p <= m -> n + (m - p) = n + m - p *)
```